Application Report
SPRA507

Sequential Addressing of I/O Ports on the TMS320C54x DSP

Clay Turner *Digital Signal Processoring Solutions*

Abstract

On the Texas Instruments (TI™) TMS320C54x, the I/O port addresses are hard-coded in the opcodes for the PORTW (I/O port write) and PORTR (I/O port read) instructions. This document discusses how the tables of data can be transferred using sequential I/O port addresses.

The fact that the I/O port addresses in the PORTR and PORTW instruction are hard-coded makes reading or writing a table of data to sequential addresses in I/O space difficult. The simplest method would be to have a sequence of PORTx instructions with each address specified explicitly. Although this method will work, it consumes as many instructions as there are entries in the table (at 2 program memory words per PORTx instruction). An alternative method takes advantage of the ability to overlay the on-chip DARAM on the C54x. A code listing is included.

Contents

Examples

Design Problem

On the TMS320C54x, the I/O port addresses are hard-coded in the opcodes for the PORTW (I/O port write) and PORTR (I/O port read) instructions. How can tables of data be transferred using sequential I/O port addresses?

Solution

The fact that the I/O port addresses in the PORTR and PORTW instruction are hard-coded makes reading or writing a table of data to sequential addresses in I/O space difficult. The simplest method is to have a sequence of PORTx instructions with each address specified explicitly. Although this method will work, it consumes as many instructions as there are entries in the table (at 2 program memory words per PORTx instruction). The alternative method shown in Example 1 takes advantage of the ability to overlay the on-chip DARAM on the C54x.

When the on-chip DARAM block is “overlaid” (due to the OVLY bit being set to 1), this block of memory is accessible from both program space and data space and mapped to the same addresses in both memory spaces. So a given memory address in program memory is physically the same location as the address in data memory. Consequently, an access to data memory can be used to modify program memory. This capability is utilized to dynamically change the address coded in the PORTx instruction so that on each pass of a loop, the address can be different. An example of this implementation is shown below in Example 1.

Example 1. Sequential Addressing of a PORTW Instruction

```
start:
             ORM   #00020h,pmst    ;set OVLY=1
             STM   #01000h,ar2     ;pointer to data memory address
             STM   #02000h,ar3     ;pointer for I/O port address
             MVMD  ar3,(portloc+1) ;update PORTW instruction
                                   ; with new address
             STM   #table_length,BRC  ;initialize BRC
             RPTB  end_block-1
portloc:     PORTW *ar2+,0h        ;copy word from data space to
                                   ; I/O space and increment
                                   ; data memory address
             MAR   *ar3+           ; increment I/O memory address
             MVMD  ar3,(portloc+1) ;update PORTW instruction
                                   ;with new address
             NOP                   ;wait for MVMD pipeline latency
             NOP                   ;wait for MVMD pipeline latency
                                   ;(portloc+1) is now updated
end_block:
```

In this example, AR2 and AR3 are used as pointers to the tables in data space and I/O space, respectively. The source table is located at address 01000h in data space. The destination table is located at address 02000h in I/O space. The MVMD instruction is used to modify the port address in the PORTW instruction. The port address is the second word of the instruction and is indicated as one address higher than the location of the PORTW instruction by using the label (portloc+1).

On each pass of the loop, data is copied from data memory address (pointed by AR2) to the I/O port address that is currently loaded in the PORTW opcode. The MAR instruction increments AR3, which keeps track of the desired I/O port address. The MVMD copies that new address into the second word of the PORTW instruction. The result is a block of locations in data memory being copied to a block of locations in I/O space.

A two-cycle latency between the MVMD instruction modifies the port address and the PORTx instruction. This occurs because the MVMD instruction writes the change to (portloc+1) in the execute phase of the pipeline, but the PORTx instruction will read the port address during the second cycle of the fetch phase of the pipeline. At least two cycles must exist between the MVMD instruction and the PORTx instruction that follows to make sure that the address has been modified before it is fetched.

The two NOP instructions in the example serve this latency, but two useful one-cycle instructions or a single two-cycle instruction could replace them. The MAR instruction could even replace one of the NOP instructions if the starting address was corrected accordingly. The port address indicated in the PORTx instruction (0h in this example) is irrelevant because it gets modified anyway. An approach similar to Example 1 can be used with the PORTR instruction. The operands of the PORTR instruction will simply be reversed.

For this implementation, each pass of the loop will require a minimum of 7 cycles to execute (assuming external memory wait states are minimized). For this code to run successfully, the OVLY bit must be set and this code must be stored in on-chip DARAM. The addresses of the data and I/O tables are not limited.